
1 Outline

• Representing binary trees in ML.

• Size, depth, and structural induction.

• Tree-based mergesort: a lesson in design and implementation.

• Specifications, correctness and proofs

• Work and span analysis

2 Trees in ML

We can use a parameterized datatype definition to introduce a type whose
values represent (binary) trees containing data from some underlying type,
such as a binary tree of integers. A binary tree is either empty, or has a root
node containing a data value (of the underlying type), and two “child” trees.
Our datatype definition is parameterized by a type variable ’a standing in for
the type for node data. This is just like the way the ML list type constructor
is parameterized by a choice of the type for list elements. In ML a “type
variable” is written like ’a or ’b.

The datatype definition

datatype ’a tree = Empty | Node of ’a tree * ’a * ’a tree

introduces a type constructor tree, together with two value constructors
Empty: ‘a tree and Node: ‘a tree * ‘a * ‘a tree -> ‘a tree for
building tree values. Since this is a user-defined type, these are the only ways
you can build tree values. Every value of type int tree is either Empty, or
has the form Node(A,x,B) where A and B are values of type int tree, and
x is an integer value. We say that A is the left-child and B is the right-child;
x is the integer “at the root”.

Example: The expression Node(Empty, 42, Node(Empty, 99, Empty))

has type int tree. This expression is also a value.
A value of type int tree represents a binary tree with integers at its

nodes; the Empty tree value contains no integer data. Every non-empty tree
has a piece of data at its root and two sub-trees or children, which may be
empty.

The constructors can also be used for pattern-matching against values of
type tree. The pattern Empty only matches the value Empty. A pattern
Node(p1, p, p2), in which p1, p and p2 are patterns, matches tree values
of the form Node(v1, v, v2) such that p1 matches v1, p matches v, and p2

matches v2. For example, the pattern Node(A, x, B) matches non-empty
tree values and binds A to the left subtree, x to the root value, and B to the
right subtree. Similarly, Node(Empty, x, Empty) matches only non-empty
trees with a single node, and binds x to the value at the root.

It is convenient to draw pictures of tree values, rather than always using
the ML syntax for tree expressions. In pictures we usually omit drawing
Empty nodes explicitly. We draw the root node at the top, subtrees lower,
left subtree to the left, and so on. (We usually don’t draw Empty explicitly,
but sometimes we may do that instead.) For example, let t be the ML
expression below:

Node(Empty, 42, Node(Empty, 9, Empty)).

This tree value T can be drawn (without showing Empty subtrees) as:

42

\

9

And the tree Node(T, 0, T) looks like:

0

/ \

42 42

\ \

9 9

In each case all the “missing” Empty subtrees can easily be filled in; their
positions are implied by the tree shape.

Some simple ML code for building “full binary trees”:

fun Leaf(x:int): int tree = Node(Empty, x, Empty);

fun Full(x:int, n:int): int tree =

if n=0 then Empty else

let val T = Full(x, n-1) in Node(T, x, T) end;

The function Leaf:int -> int tree builds a tree with a single node. We
refer to this as a “leaf”. The expression Full(2,5) evaluates to a “full”
binary tree with 2 at each node and with depth (or height) 5.

Draw pictures of Leaf 42 and Full(42,3). Note that the expression
Full(42,3) evaluates to a tree value with 7 nodes, each with the integer 42.

Evaluation and equality

To evaluate the ML expression Node(e1, e, e2) we must evaluate e1 to a
tree value (say v1), evaluate x to an integer (say v), and evaluate e2 to a
tree value (say v2). The result is the tree value Node(v1, v, v2).

Two ML expressions of type int tree are equal if they evaluate to equal
tree values, or they both fail to terminate.

For example, Leaf 42 and Full(42, 1) are equal, because they both
evaluate to Node(Empty, 42, Empty).

The type int tree defined as above is an ML equality type (because int

is an equality type), so we can use ML = for testing when two integer stree
values are identical. We won’t use this feature in our sorting functions, but
it may be handy for testing.

Structural induction for trees

To reason about trees, and to define and reason about functions on trees, we
need a form of induction that works with trees. The structure of the datatype
definition for trees is the key here. Every tree value is either Empty, or has
the form Node(l,x,r), where l and r are tree values and x is a value. This
amounts to an inductive way to generate tree values. Initially we start with
the tree value Empty. Then we generate all trees obtainable by applying the
Node constructor to already existing trees and some integer. And we keep
repeating this process. At each stage the set of trees generated so far grows
larger.

We can prove a property like “for all trees T, P (T) holds”, as follows:

(i) Base case: Show that P (Empty) holds.

(ii) Inductive step: Assume as Induction Hypothesis that l and r are
tree values such that P (l) and P (r) hold; show that for all values
x, P (Node(l, x, r)) holds.

(iii) It follows from (i) and (ii) that P (T) holds, for all tree values T.

This proof method is called structural induction for trees.
There is also a corresponding principle of structural recursion for function

definitions. To define a function F on all tree values:

• Give a clause defining F (Empty).

• Give a clause defining F (Node(A, x, B)) in terms of x, F (A) and F (B).

Such clauses are sufficient to completely specify the intended value of F (T),
for all tree values T. We say that these clauses constitute a definition (of F)
by structural recursion on trees.

For every datatype definition in ML there is an analogous principle of
structural induction. We will see many examples in lab and later in the
semester. You have already seen one: the kind of list induction discussed
earlier is basically a form of structural induction, since the ML list types are
defined in terms of nil and “cons”: every value of type t list is either nil
or has form v::L, where v is a value of type t and L is a value of type t list.

size and depth of trees

The size of a tree is the number of nodes it contains. So the size of Empty
is 0, and the size of a non-empty tree is the sum of 1 and the sizes of its
two children. We can easily define an ML function size : ‘a tree -> int

that computes the size of a tree, using structural recursion:

fun size Empty = 0

| size (Node(t1, _, t2)) = size t1 + size t2 + 1

It is easy to check that

size(Full(42,3)) = 7

Intuitively, size(t) is the number of nodes in t; using structural induction,
it is easy to prove this.

The depth of a tree is the length of the longest path from the root of the
tree to an Empty subtree. A path is a sequence of nodes. The depth of an
empty tree is defined to be 0.

For all trees t, size(t) ≥ 0 and depth(t) ≥ 0; and if t′ is a child of t,
then depth t′ < depth t and size t′ < size t. So we can also use induction
on tree depth, or induction on tree size, as techniques for proving properties
of trees, or of functions operating on trees.

We can define depth : ’a tree -> int very easily in ML, using the
built-in function Int.max : int * int -> int as follows:

fun depth Empty = 0

| depth (Node(t1, _, t2)) = 1 + Int.max(depth t1, depth t2)

ASIDE: structural induction on trees, induction on tree size, induction
on tree depth, as well as simple and complete induction on non-negative
integers, are all special cases of a general technique known as well-founded
induction. Perhaps you have seen well-founded orderings in action in some
other math or CS classes.

In-order traversal

Here is a function that builds a list of integers from an integer tree, by
making an in-order traversal of the tree, collecting data into a list. In-order
traversal of a non-empty tree involves traversing the left-child, then the root,
and then traversing the right-child; we also use in-order traversal on the
sub-trees. Obviously this description suggests that we define a structurally
recursive function!

This function is used mainly in specifications, but serves as an example of
how to define a function that operates on trees: use clauses, one for the empty
tree and one for non-empty trees, using pattern-matching to give names to
the components of a tree.

(* trav : ’a tree -> ’a list *)

fun trav Empty = [ ]

| trav (Node(t1, x, t2)) = trav t1 @ (x :: trav t2);

(* Specification: *)

(* trav t = a list of the data in the tree t, *)

(* as seen in an in-order traversal of the tree *)

(* Say "x is in t" if x is a member of the list trav(t). *)

For example, for the tree t above, trav(t)=[42,9]. And

trav(Node(t, 0, t)) = trav t @ (0 :: trav t)

= [42, 9] @ (0 :: [42, 9])

= [42, 9, 0, 42, 9].

We prove, by structural induction on trees, that for all trees t, trav(t)

evaluates to a list of length equal to size(t). This is the same as saying
“trav(t) = a list of length size(t)”. Proof: by structural induction
on t.

• Base case: For t = Empty. Since size(Empty) = 0 we must show that
trav Empty = [ ]. This is obvious from the function definition.

• Inductive case: For t = Node(t1, x, t2). Let n1 = size t1 and n2

= size t2. So size(t) is n1+n2+1. Assume as Ind. Hyp. that

(i) trav t1 = a list (say L1) of length size(t1)

(ii) trav t2 = a list (say L2) of length size(t2).

Then by definition of trav we have

trav (Node(t1, x, t2))

= (trav t1) @ (x :: trav t2)

= L1 @ (x :: L2) by (i) and (ii)

= a list of length n1 + 1 + n2

= a list of length size(Node(t1, x, t2))

as needed.

3 Sorting trees of integers

Let’s briefly review what we said about integer lists and sorting.
As before, we refer to:

datatype order = LESS | EQUAL | GREATER;

(* compare : int * int -> order *)

fun compare(x:int, y:int):order =

if x<y then LESS else

if y<x then GREATER else EQUAL;

(* compare(x,y)=LESS if x<y *)

(* compare(x,y)=EQUAL if x=y *)

(* compare(x,y)=GREATER if x>y *)

A list of integers is sorted if each item in the list is ≤ all items that occur
later in the list.

We will also refer to the ins function, used as a helper when we did
insertion sort on lists of integers.

(* ins : int * int list -> int list *)

(*REQUIRES L is sorted *)

(* ENSURES ins(x, L) evaluates to a sorted permutation of x::L. *)

fun ins (x, [ ]) = [x]

| ins (x, y::L) = case compare(x, y) of

GREATER => y::ins(x, L)

| _ => x::y::L

When is a tree sorted?

We adapt the list-based definition to the arboreal setting in a fairly obvious
way. An integer tree is said to be sorted if at each node, the integer at that
node is ≥ all integers in the left subtree and ≤ the integers in the right
subtree. Here is an inductive way to characterize sortedness:

• The tree Empty is sorted.

• A non-empty tree value Node(A,x,B) is sorted if and only if every
integer in A is ≤ x, every integer in B is ≥ x, and A and B are sorted.

For brevity we may sometimes write A ≤ x ≤ B, with A and B being tree
values and x an integer. While it is possible to write an ML function for
testing if a tree is sorted or not, we will not bother to do so here. We won’t
ever need to check for sortedness in implementing an algorithm for sorting
trees; instead we’ll design our code so that it is guaranteed to return a sorted
tree, even without checking.

How to sort a tree

Our algorithm for sorting an integer tree is described informally as follows:

• If the tree is empty, do nothing.

• Otherwise, (recursively) sort the two subtrees; merge the sorted sub-
trees into a single sorted tree; then insert the root value into its correct
position.

This suggests that we design helper functions for inserting an item into a
sorted tree, and merging two sorted trees into one. Later we will see that
the merging operation itself needs a helper function, for splitting a tree into
two subtrees, using a given integer value to determine which items from the
tree go into the first or second subtree. Also it will be important to think
carefully about what assumptions it is safe to make about the arguments to
be supplied to these helper functions.

Insertion for trees

The tree-based analogue of the insertion function on lists turns out to be
just what we need, a truly helpful function in the code that follows. We use
capitalization to distinguish this function from the ins function on lists.

(* Ins : int * int tree -> int tree *)

(* REQUIRES T is a sorted tree *)

(* ENSURES Ins(x, T) = a sorted tree consisting of x and all of T *)

fun Ins (x, Empty) = Node(Empty, x, Empty)

| Ins (x, Node(t1, y, t2)) =

case compare(x,y) of

GREATER => Node(t1, y, Ins(x, t2))

| _ => Node(Ins(x, t1), y, t2)

Compare this code with the code for ins. See how similar it is. We now show
how to prove by structural induction that Ins satisfies this specification. Let
P (T) be the property that

For all integer values x, if T is sorted, then Ins(x,T) evaluates to
a sorted tree consisting of x and the items of T.

We prove “For all tree values T, P (T) holds”, by structural induction on T.

• The base case is simple. P (Empty) holds, because Empty is sorted and
Ins(x, Empty) evaluates to Node(Empty, x, Empty). This is a tree
value (because x is an integer value by assumption, and Empty is a tree
value), is obviously sorted, and consists of just x, as required.

• For the inductive step we argue as follows. Suppose t1 and t2 are tree
values such that P (t1) and P (t2) hold, and let y be an integer value.
We show that P (Node(t1, y, t2)) holds. To do this, let x be an integer
value and suppose Node(t1, y, t2) is sorted. We must show that
Ins(x, Node(t1, y, t2)) evaluates to a sorted tree value consisting
of x and all items of Node(t1, x, t2). We also know that t1 and t2

are sorted, and t1 ≤ y ≤ t2. From the definition of Ins we see that

(a) Either x>y, in which case we have

Ins(x, Node(t1, y, t2)) =>* Node(t1, y, Ins(x, t2)).

By induction hypothesis P (t2), Ins(x,t2) evaluates to a sorted
tree (say u2) consisting of x and all of t2, so we get

Ins(x, Node(t1, y, t2)) =>* Node(t1, y, u2),

t1 ≤ y ≤ u2, and t1 and u2 are sorted tree values, so this is a
sorted tree value. Clearly it consists of x and y and all of t1 and
t2. Thus P (Node(t1, y, t2)) holds in this case.

(b) Or x ≤ y, in which case we have

Ins(x, Node(t1, y, t2)) =>* Node(Ins(x, t1), y, t2).

By induction hypothesis P (t1), Ins(x,t1) evaluates to a sorted
tree (say u1) consisting of x and all of t1, so we get

Ins(x, Node(t1, y, t2)) =>* Node(u1, y, t2),

u1 ≤ y ≤ t2, and u1 and t2 are sorted tree values, so this is a
sorted tree value. Clearly it consists of x and y and all of t1 and
t2. Thus P (Node(t1, y, t2)) holds in this case also.

So P (Node(t1, y, t2)) holds, as needed for the inductive step.

The above specification and proof use evaluational notation and evaluational
reasoning. One can also state and prove an equational specification for Ins,
using “value equations” derived from the function definition. It follows from
the function definition that for all integer values x and y, and all tree values
t1 and t2, the following equations hold:

Ins(x, Empty) = Node(Empty, x, Empty)

Ins(x, Node(t1, y, t2)) = Node(t1, y, Ins(x, t2)) if x>y

Ins(x, Node(t1, y, t2)) = Node(Ins(x, t1), y, t2) if x<y or x=y

We can prove that

For all integer values x and all sorted tree values T, there is a
sorted tree value S such that

Ins(x, T) = S

and S consists of x and all the integers in T.

To prove this result you can basically adapt the evaluational proof steps from
above and write a corresponding equational proof.

Splitting a sorted tree

In adapting the mergesort algorithm to operate on trees we need a suitable
analog to the split function. (In class we motivated this need by trying to
figure out how to merge two sorted trees. At some point we found ourselves
wanting to split a tree into two trees.) It isn’t easy to figure out a good
way to hew a tree into two roughly equal sized pieces, based solely on the
structure of the tree (by analogy with the way the split function on lists
worked). Instead, we will start from a tree and an integer, and break the
tree into two trees, one consisting of items less-or-equal to the integer and the
other consisting of items greater than or equal to the integer. (There is some

wiggle room here concerning where the items equal to this integer should
go.) We will only ever need to use this method on a sorted tree, as you will
observe when we develop the code. We also design the function so that when
applied to a sorted tree it produces a pair of sorted trees. Indeed the design
of this function takes advantage of the assumption that the tree is already
sorted, a fact that we echo in the way we write the function’s specification.

(* SplitAt : int * int tree -> int tree * int tree *)

(* REQUIRES T is sorted *)

(* ENSURES SplitAt(x, T) = (A,B) where

A and B are sorted trees, A <= x <= B,

A and B consist of the items from T *)

fun SplitAt(x, Empty) = (Empty, Empty)

| SplitAt(x, Node(t1, y, t2)) =

case compare(y, x) of

GREATER => let

val (l1, r1) = SplitAt(x, t1)

in

(l1, Node(r1, y, t2))

end

| _ => let

val (l2, r2) = SplitAt(x, t2)

in

(Node(t1, y, l2), r2)

end

This function is structurally inductive, because in the recursive clause
SplitAt(x,Node(t1,y,t2)) either calls SplitAt(x,t1) or SplitAt(x,t2),
in each case making a recursive call on a subtree. We prove that SplitAt

satisfies this specification, by structural induction. To be precise, we prove
“For all tree values T, P (T)”, where P (T) is the property that if T is sorted,
then for all values x, SplitAt(x, T) is equal to a pair of sorted trees (A,B)
such that A ≤ x ≤ B and A, B consist of the items of T.

• Base case: Empty is sorted and has no elements, so we need to show that
for all values x, SplitAt(x,Empty) is equal to a pair of sorted trees with
no elements. By definition we have SplitAt(x, Empty)=(Empty,Empty),
and the requirements hold trivially.

• Inductive step: Let t be a sorted tree of form Node(t1, y, t2) and
assume that SplitAt satisfies the spec on t1 and on t2. By assumption
that the whole tree is sorted, we also know that t1 and t2 are sorted,
and that t1 ≤ y ≤ t2. We show that SplitAt(x,t) is equal to a pair
of sorted trees with the required properties. There are two sub-cases
to analyze, branching on the result of comparing the values of x and y.

(a) If y > x we have

SplitAt(x,t) = (l1, Node(r1,y,t2))

where (l1, r1) = SplitAt(x, t1). By Induction Hypothesis,
l1 ≤ x ≤ r1 and l1, r1 are sorted trees consisting of the items
from t1. So l1 ≤ x ≤ Node(r1, y, t2) and r1 ≤ y ≤ t2. And
Node(r1, y, l2) is a sorted tree consisting of the items from t1,
y and l2. Together with l1 this covers all the items from t.

(b) If y ≤ x we have

SplitAt(x,t) = (Node(t1, y, l2), r2)

where (l2, r2) = SplitAt(x, t2).

By induction hypothesis, l2 ≤ x ≤ r2 and l2, r2 are sorted trees
comprising the items from t2. Hence t1 ≤ y ≤ l2 and Node(t1, y, l2)
is sorted. Also Node(t1, y, l2) ≤ x ≤ l2. (Fill in the remaining details.)

That completes the proof.

Merging two sorted trees

Now the tree-based analog of merge: a function that takes a pair of sorted
trees and combines their data into a single (also sorted) tree. We use SplitAt
as a helper.

(* Merge : int tree * int tree -> int tree *)

(* REQUIRES t1 and t2 are sorted trees *)

(* ENSURES Merge(t1, t2) = a sorted tree

consisting of the items from t1 and t2 *)

fun Merge (Empty, t2) = t2

| Merge (Node(l1,x,r1), t2) = let

val (l2,r2) = SplitAt(x,t2)

in

Node(Merge(l1,l2), x, Merge(r1,r2))

end

The proof that Merge meets this spec relies on the fact that SplitAt meets
its own specification. Indeed, we deliberately chose a spec for SplitAt that
would help us to prove Merge correct. That’s one of the skills that we want
you to learn: the art of choosing helper functions and specs wisely! We claim
that “For all sorted tree values T, P (T) holds”, where P (T) is the property
that

For all sorted tree values U, Merge(T, U) is equal to a sorted tree
value comprising the items from T and U.

The proof is by structural induction on T. (Since sorted trees have sorted
children, it is perfectly OK to do this kind of structural induction on sorted
trees!)

• P (Empty) holds obviously. (Fill in the details.)

• For the inductive case, suppose l1 and r1 are sorted tree values for
which P (l1) and P (r1) hold. Let x be an integer value and assume
Node(l1, x, r1) is sorted. We must show that P (Node(l1, x, r1))
holds. Let U be a sorted tree value. By definition of Merge we have

Merge(Node(l1, x, r1), U) = Node(Merge(l1, l2), x, Merge(r1, r2))

where (l2, r2) = SplitAt(x, U). By the proven spec for SplitAt

(which is applicable here because U is a sorted tree), l2 ≤ x ≤ r2 and
l2, r2 are sorted and comprise the items from U. By assumption that
the original tree is sorted, we have l1 ≤ x ≤ r1, and l1, r1 are sorted.
So by Induction Hypothesis, there are sorted tree values l and r such
that

Merge(l1, l2) = l, Merge(r1, r2) = r

and l comprises the items from l1, l2, and r comprises the items from
r1 and r2. Hence l ≤ x ≤ r and Node(l, x, r) is a sorted tree,
consisting of the items from l1, x, r1, U, as required.

That completes the proof. (We skipped over a few details – make sure you
understand how to fill in the gaps.)

Lesson

It’s important to notice how in the above code analysis the specs for the
various helper functions play a crucial rôle. Just in the nick of time, it
turned out we could appeal to an induction hypothesis, which was applicable
because we had shown that one of the helper functions behaved well. We
were careful to only use a helper function with arguments that satisfy the
requirements of the helper spec, and in the proof details we confirmed that
the guarantees made by the helper functions (when used in this manner)
were sufficient to ensure that the rest of the code meets its own spec. If we
hadn’t shown that SplitAt preserves sortedness, we would have no basis for
claiming that Merge preserves sortedness. The lesson is: choose your helper
functions wisely, choose their specs wisely (with an eye to how you will use
them), and nail down the correctness proof (at least with a sketch of the key
details).

The tree-sorting function Msort

Using Ins and Merge, and guided by their (proven) specs, we are now ready
to define a mergesorting function for integer trees. The hard work has already
been done; now comes the easy and more immediately rewarding part! The
tree sorting function in ML is defined in a way that mimics the algorithm we
sketched earlier.

(* Msort :int tree ->int tree *)

(* REQUIRES T is a value of type tree *)

(* ENSURES Msort(T) = a sorted tree

consisting of the items of T *)

fun Msort Empty = Empty

| Msort (Node(t1, x, t2)) = Ins (x, Merge(Msort t1, Msort t2))

Again the proof that Msort meets this spec uses the facts (shown earlier)
that Ins and Merge satisfy their specs. And again these helper specs were
carefully chosen to make this all fit together! Exercise: fill in the proof
details. Contrast with the proof given in the earlier lecture notes for the
mergesort function on lists.

4 Exploration

To illustrate how the various functions discussed above actually work on a
specific tree example, try running the following code and drawing pictures of
the tree values produced in each stage.

val T1 = Node(Leaf 3, 6, Leaf 1);

val T2 = Node(Leaf 5, 2, Empty);

val T = Node(T1, 4, T2);

val M1 = Msort T1;

val M2 = Msort T2;

val M = Merge(M1, M2);

val S = Ins(4, M);

You might also want to try some examples using SplitAt. Furthermore, it
may be useful to define some functions for extracting a list containing the
integers in a tree. In lab you will discuss traversal lists and in-order, pre-order
and post-order traversal of trees. It turns out that an integer tree is sorted
if and only if its in-order traversal list is a sorted list of integers. So one can
easily check if a tree is sorted by looking at its in-order traversal list. Just
for some irrelevant fun(?), try to figure out a decent specification of what
SplitAt, Ins and Merge do when applied to arguments that do NOT satisfy
the REQUIRES assumptions used above.

5 Comments on sorting trees

For an integer list L there is just one sorted list that contains the same items
as L. So it makes sense to talk about “computing the sorted version of L”. In
contrast, for a collection of at least 2 integers there can be multiple different
trees containing the same integers. Indeed, there can be many different sorted
trees containing the same integers. So the specifications and proofs so far
don’t really tell us much about the shapes of the trees produced by sorting.
It would be nice if we guaranteed to produce balanced trees, in which at each
node the numbers of integers in the two child subtrees differ by at most 1.
However, even if we start with a balanced (unsorted) tree, the functions that
we have defined so far do not always produce balanced results. We will return
to this point shortly.

6 Size analysis

We can prove some fairly obvious facts about the effects of the operations on
the size of a tree.

(1) For all trees t and integers x,

size(Ins(x, t)) = size(t) + 1.

(2) For all trees t and integers y, if SplitAt(y,t)=(t1, t2) then

size(t1) + size(t2) = size(t).

(3) For all trees t1 and t2,

size(Merge(t1, t2)) = size t1 + size t2.

(3) For all trees t,

size(Msort t) = size t.

In each case, you can prove the result by structural induction. Check that
these results are consistent with the examples from before.

7 Depth analysis

We can prove some useful (and intuitively obvious) results about depth.
These will be helpful when we analyze the runtime behavior of the code. The
following results are provable, by choosing an appropriate kind of induction.

(1) For all trees t and integers x,

depth(Ins(x, t)) <= depth(t) + 1.

[Use structural induction, since Ins(x, t) makes a recursive call
on a child of t.]

(2) For all trees t and integers y, if SplitAt(y,t)=(t1, t2) then

depth(t1) <= depth(t) and depth(t2) <= depth(t).

17



[Use structural induction, since SplitAt(y,t) makes a recursive
call on a child of t.]

(3) For all trees t1 and t2,

depth(Merge(t1, t2)) <= depth t1 + depth t2.

[Use induction on the structure of t1.]

(3) For all trees t,

depth(Msort t) <= depth t.

[Use induction on the structure of t.]

Check that these results are consistent with the examples from earlier.

8 Work and span

We’ve shown how to derive recurrence relations for the work of a sequentially
executed piece of code, and how to estimate asymptotically what the runtime
is on “‘large” inputs, using big-O notation.

Now we have some functions operating on trees for which it makes a lot
of sense to consider using parallel evaluation. The span of a code fragment is
obtained by assuming that we have as many parallel processors as we need,
and taking the maximum runtime of code pieces that can be evaluated inde-
pendently; we still use addition for the run ties of code fragments that need
to be executed in sequential order, typically because of a data dependency:
one fragment needs the result of the other. Operating on trees allows us
in principle to sort the left and right children of a node in parallel, since
their results do not depend on each other. Of course, these tasks need to be
completed before the merging phase. And the splitting phase needs to go
first.

These facts guide us in analyzing the span. Here is a rough outline. With
trees there are two “largeness” measures of interest: depth and size.

• The work and span for Ins(x,t) is O(d), where d is the depth of t.
Reason: Ins(x, t) makes a single recursive call, on a subtree with depth
decreased by 1.

• SplitAt(y, t) has span O(d), where d = depth t. Reason: makes a
single recursive call, on a tree with depth one less.

• Merge(t1, t2) has span O(d1d2), where d1, d2 are depth t1, depth t2.

• Assuming that the trees produced by Msort are balanced, so that their
depth is about the logarithm of their size, Msort(t) has span O(d3),
where d is the depth of t. Reason: making the balance assumption
leads us to the recurrence

SMsort(d) = SIns(d) + SMerge(d− 1) + SMsort(d− 1)
= d + (d− 1)2 + SMsort(d− 1)

for balanced trees of depth d > 1. Expanding out, and observing that
the sum of the first d squares is proportional to d3, we deduce that
the span is O(d3). Since the size n of a balanced tree and its depth d
satisfy d = O(log n), our analysis shows that the span for Msort(t) on
balanced trees of size n is O((log n)3).

Thus (ignoring constants), when we sort a billion integers in a balanced
tree, the length of the longest critical path is about 27000 operations, so we
can exploit over a million processors!

This would be true, except for the bug in the above analysis! We assumed
implicitly in the rough analysis (and explicitly in the preamble) that the trees
passed by Msort to Merge were balanced. However, this is not necessarily
the case, because even if we assumed that the original tree was balanced,
these two trees have been built by calling Msort (albeit on balanced trees).
We haven’t proven that Msort applied to a balanced tree will produce a
balanced tree. In fact, this isn’t necessarily true. The best that our analysis
really predicts is that the span of this algorithm can’t actually be better than
this bound, because we obtained this bound by making the most optimistic
assumptions about the structure of the tree.

Later we will discuss how to implement binary trees with insertion and
deletion operations that are guaranteed to build trees with a reasonable bal-
ance property built in. When we get there, you might want to come back and
see how you could adapt the code above to fit with these better behaved trees.

(Look again at the examples from before: are the various trees constructed
there balanced?)

Exercise
A student pointed out that there is a way to define a version of tree-mergesort
that avoids using Ins, instead calling Merge:

fun Msort’ Empty = Empty

| Msort’ (Node(t1, x, t2)) =

Merge(Node(Empty,x,Empty), Merge(Msort’ t1, Msort’ t2))

(i) Would this be extensionally equivalent to the previous function?
(How would one prove it?)

(ii) Does this function satisfy the same specification as before, i.e. does it
still sort?

(iii) And is it as efficient, or more efficient, asymptotically?
(How could you determine this?)

9 Self-test 7

1. Write an ML function leaves : int tree -> int list such that for
all values T, leaves(T) = the list of the integers occuring at leaf nodes
of T. For example, leaves(Full(42,3))=[42,42,42,42].

2. Let T be a tree with depth d. Why is the size of T at most 2d − 1?

3. Write an ML function treesum : int tree -> int such that for all
values T, treesum(T) evaluates to the sum of the integers at the nodes
of T. For example, treesum(Full(42,3)) should evaluate to 294.

4. Wite an Ml function leafsum : int tree -> int such that for all
values T, leafsum(T) evaluates to the sum of the integers at leaf nodes
of T. We interpret this quantity as 0 if the tree is Empty.
Do not use leaves from above!

5. State and prove a theorem about the value of treesum(Full(x,n))

when n is a non-negative integer.

6. How many different tree values T have exactly three nodes containing
the integers 1,2,3? Of these, how many are sorted trees?

7. Define an ML function balanced : ‘a tree -> bool such that for all
tree values T, balanced(T)=true if at each node of T the sizes of the left
and right children differ by at most 1. Otherwise balanced(T)=false.

8. Calculate the value produced by the following piece of code:

val T1 = Node(Leaf 1, 6, Leaf 3);

val T2 = Node(Empty, 5, Leaf 2);

val T = Node(T1, 4, T2);

Msort T;

9. Prove that Msort satisfies its specification. You can assume given
proofs that Merge and Ins satisfy their specifications.

10. Let Tn be the tree value of the expression Full(42, n), for n ≥ 0. This
is a full binary tree of depth n with 42 at each node. State and prove
an assertion about the (shape of the) value of Msort(Tn).

